

# Excel Hacks

Formeln und Funktionen effektiv einsetzen

» Hier geht's  
direkt  
zum Buch

# DIE LESEPROBE

# Datum und Uhrzeit

## 1.1 Zeitberechnungen

### 1.1.1 Eine aktualisierbare Uhrzeit einfügen

#### Problem

In einem Formular soll die aktuelle Zeit angezeigt werden. Die Zeit soll sich beim Öffnen der Datei oder beim Drücken der Taste **[F9]** automatisch aktualisieren.

	A	B
1	<b>Eine aktualisierbare Zeit</b>	
2		Formel:
3	23.04.22 17:07	=JETZT()
4	17:07:39	=JETZT()
5	17:07	=JETZT()
6	letzte Aktualisierung um 17:07:39	= "letzte Aktualisierung um "&TEXT(JETZT());"HH:MM:SS")
7		

#### Lösung

Schreiben Sie die Funktion = JETZT() in die Zelle, in der die aktuelle Uhrzeit angezeigt werden soll.

#### Hintergrund

Die Formel wird bei jeder Neuberechnung der Arbeitsmappe aktualisiert.

Wählen Sie zusätzlich **START|FORMAT|ZELLEN**, Registerkarte **ZAHLEN**, Kategorie **UHRZEIT** und dann das gewünschte Zeitformat.

Diese Funktion kann auch in Text eingebettet werden, wie die Formel in folgendem Beispiel zeigt:

```
= "letzte Aktualisierung um "&TEXT(JETZT());"HH:MM:SS")
```

#### Verwendete Funktionen

JETZT()

TEXT(Wert;Format)

## 1.1.2 Zahlen in Minuten umwandeln

### Problem

Sie möchten Minutenangaben, die als reine Zahlenwerte eingegeben wurden, in einen Zeitwert mit Stunden und Minuten umwandeln. In der Zelle A3 steht beispielsweise der Wert 175 für eine geleistete Arbeitszeit von 175 Minuten. Dieser Wert soll zu einem echten Zeitwert umgerechnet werden, sodass in der Zelle C3 die Uhrzeit »02:55:00« steht.

C3			<b>f x Σ v =</b>	=ZEIT(0;A3;0)	
	A	B	C	D	E
1	Zahl	in	Minuten	umwandeln	
2				Formel:	
3	175		02:55:00	=ZEIT(0;A3;0)	
4					

### Lösung

Wenn nur Zahlenwerte für die Minuten (und eventuell auch für Stunden und Sekunden) in der Zelle stehen, können Sie diese direkt mit der Tabellenfunktion ZEIT() in einen »echten« Zeitwert umrechnen. Dazu schreiben Sie in die Zelle C3 die Formel:

```
=ZEIT(0;A3;0)
```

Als Ergebnis erhalten Sie die Zahl: »0,121527777777778«, formatiert mit dem Zahlenformat HH:MM:SS die Zeit »02:55:00«. Sie können aber auch jedes andere verfügbare Zeitformat verwenden.

### Hintergrund

Die Funktion ZEIT() erwartet drei Parametereingaben: Stunden, Minuten und Sekunden. Fehlt Ihnen eine Angabe, setzen Sie den entsprechenden Parameter einfach auf 0.

### Verwendete Funktionen

ZEIT(Stunde;Minute;Sekunde)

## 1.1.3 Wie errechnen Sie aus Minuten die Stunden und Minuten als Dezimalzahl?

### Problem

Die geleistete Arbeitszeit wird nur in Minuten angezeigt. Sie möchten aus dieser Zahl die Stunden und die Minuten als Dezimalzahl berechnen.

B8			
=GANZZAHL(A8/60) + REST((A8/60-GANZZAHL(A8/60));60)			
	A	B	C
1	Wie errechne ich aus Minuten die Stunden und die Minuten		
2	Minuten	Std. und Min.	Formel:
3	330	05:30:00	=ZEIT(GANZZAHL(A3/60);REST(A3/60);0)
4	60	01:00:00	=ZEIT(GANZZAHL(A4/60);REST(A4/60);0)
5			
6	Stunden und Minuten als Dezimalzahl		
7	Minuten	Stunden und Minuten als Dezimalzahl	
8	330	5,50	=GANZZAHL(A8/60) + REST((A8/60-GANZZAHL(A8/60));60)
9	345	5,75	=GANZZAHL(A9/60) + REST((A9/60-GANZZAHL(A9/60));60)
10			

### Lösung

In der Zelle A3 steht die Arbeitszeit in Minuten: »330«.

In die Zelle B3 schreiben Sie die Formel:

```
=ZEIT(GANZZAHL(A3/60);REST(A3;60);0)
```

Als Ergebnis erhalten Sie die Dauer: »05:30«.

Soll dieses Ergebnis als Dezimalzahl ausgegeben werden, mit der Sie auch weiterrechnen können, verwenden Sie die Formel:

```
=GANZZAHL(A8/60) + REST((A8/60-GANZZAHL(A8/60));60)
```

Die Formel geht davon aus, dass die umzurechnenden Minuten in der Zelle A8 stehen. Als Ergebnis erhalten Sie »5,50« (Stunden), was auch zu weiteren Berechnungen verwendet werden kann.

### Hintergrund

Diese Formel errechnet zuerst mithilfe der Funktion GANZZAHL() die Stunden aus den übergebenen Minuten und addiert anschließend mithilfe der Funktion REST() die verbleibenden Minuten als Dezimalzahl.

### Verwendete Funktionen

GANZZAHL()

REST()

ZEIT()

## 1.1.4 Uhrzeiten runden

### Problem

Wie können Sie eine Uhrzeit im Format HH:MM:SS auf ganze Minuten runden?

## Kapitel 1

### Datum und Uhrzeit

B3			=RUNDEN(A3*1440;0)/1440	
	A	B	C	D
1	<b>Uhrzeiten runden</b>			
2	<b>Uhrzeit:</b>	<b>auf Min. gerundet:</b>	<b>Formel:</b>	
3	07:15:25	07:15	=RUNDEN(A3*1440;0)/1440	
4	07:15:31	07:16		
5	07:35:45	07:36		
6				
7		<b>auf Std. gerundet</b>		
8	07:30:00	08:00	=RUNDEN(A8*24;0)/24	
9	07:29:15	07:00		
10				
11		<b>auf 5 Min. gerundet</b>		
12	07:32:00	07:30	=RUNDEN(A12*24/(1/12);0)*(1/12)/24	
13	07:36:00	07:35		
14				
15		<b>auf 10 Min. gerundet</b>		
16	07:32:00	07:30	=RUNDEN(A12*24/(1/6);0)*(1/6)/24	
17	07:36:00	07:40		
18				
19		<b>auf 15 Min. gerundet</b>		
20	07:32:00	07:30	=RUNDEN(A20*24/(1/4);0)*(1/4)/24	
21	07:38:00	07:45		
22				
23	<b>Dezimalwerte in Stundenangaben umrechnen</b>			
24	<b>Uhrzeit:</b>			
25	6,36	06:21:36	=A25/24	
26				
27		<b>auf Min. gerundet:</b>		
28	6,36	06:22:00	=RUNDEN(A28*60;0)/1440	
29				

### Lösung

In der Zelle A3 steht die Uhrzeit, beispielsweise »07:15:25«.

In die Zelle B3 schreiben Sie die Formel:

```
=RUNDEN(A3*1440;0)/1440
```

### Hintergrund

Zuerst wird die Uhrzeit in Tagesminuten umgerechnet (24 Stunden mal 60 Minuten = 1440 Minuten). Nach der Rundung erfolgt eine Division durch die Zahl 1440, um den Zeitwert wieder in Stunden und Minuten umzuwandeln. Soll die Uhrzeit auf ganze Stunden gerundet werden, brauchen Sie die Formel nur abzuwandeln in:

```
=RUNDEN(A3*24;0)/24
```

### Verwendete Funktion

```
RUNDEN(Zahl [, Stellen])
```

## 1.1.5 Mit Uhrzeiten rechnen (Stundenlohn – Tageslohn – Durchschnittsgeschwindigkeit)

### Problem

Wie können Sie den Stundenlohn aus dem bezahlten Lohn sowie den geleisteten Arbeitsstunden errechnen?

Mit Uhrzeiten rechnen (Stundenlohn-Tageslohn-Durchschnittsgeschwindigkeit)							
Stundenlohn berechnen				Durchschnittsgeschwindigkeit			
Arbeitslohn:	Arbeitszeit:	Stundenlohn:		Kilometer:	Fahrzeit:	Durchschnittsgeschwindigkeit:	
560,00 €	08:15	67,88 €	=A5/(B5*24)	530 km	04:15:20	124,54 km/h	=F5/(G5*24)
Tageslohn berechnen							
Arbeitsbeginn	Arbeitsende	Arbeitszeit:	Stundenlohn	Tageslohn	Formel:		
06:15:00	15:30:00	09:15:00	10,00 €	92,50 €	=(B9-A9)*24*D9		

### Lösung

Schreiben Sie in die Zelle A5 den erhaltenen Lohn: 560,00 €.

In die Zelle B5 tragen Sie die Stundenzahl ein: 8:15.

In die Ergebniszelle C5 schreiben Sie folgende Formel:

```
=A5/(B5*24)
```

Als Ergebnis erhalten Sie: »67,88 €«.

### Hintergrund

Um ein richtiges Ergebnis zu erhalten, müssen Sie die Dezimalzahl, die hinter der Stundenzahl steht, erst durch die Multiplikation mit 24 auf eine 24-Stunden-Basis bringen. Mit der so gewonnenen Dezimalzahl kann nun der Stundenlohn errechnet werden. Steht in A1 »12:00« Uhr und in A2 »18:00« Uhr, so können Sie mit der Formel =A2-A1 die Zeitdifferenz bestimmen. Sie erhalten als Ergebnis »06:00«.

Intern rechnen Calc und Excel grundsätzlich ohne Zahlenformate, also mit Dezimalwerten. Dabei gibt es für jede Uhrzeit zwischen 0:00 Uhr und 24:00 Uhr einen Dezimalwert zwischen 0 und 1. Im obigen Beispiel rechnet das Programm dabei ganz einfach: 0,75 - 0,5 = 0,25. Dieses Ergebnis entspricht der formatierten Zeit: »06:00 Uhr«.

Auf der Basis von Dezimalzahlen lässt sich jede beliebige Uhrzeit bis auf die Sekunde genau abbilden. Soll umgekehrt eine Dezimalzahl in eine Uhrzeit verwandelt werden, teilen Sie einfach die Dezimalzahl durch 24. Zum Beispiel liefert die Division von =7,5/24 das Ergebnis »07:30« (mit dem Zahlenformat HH:SS formatiert).

**Tipp**

In gleicher Weise können Sie die Kilometerzahl durch eine Zeitangabe teilen, sodass als Ergebnis die Durchschnittsgeschwindigkeit errechnet wird. Die Anzahl der Kilometer geteilt durch die so ermittelte Dezimalzahl der Zeit ergibt die Durchschnittsgeschwindigkeit. Das Ergebnis wird mit dem benutzerdefinierten Zahlenformat 0,00" km/h" angezeigt.

**1.1.6 Positive und negative Zeitdifferenzen darstellen**

**Problem**

Bei einem Trainingslauf war als voraussichtliche Ankunftszeit 15:00 Uhr vorgegeben. Die tatsächliche Ankunftszeit wurde erfasst und soll in Calc oder Excel ausgewertet und dargestellt werden. Wie können Sie die Zeitdifferenzen ausweisen?

Positive und negative Zeitdifferenzen				
geplante Ankunftszeit	tatsächliche Ankunft:	Differenz	Formel:	
15:00	14:55	-00:05	=WENN(B3-A3<0;"-"&TEXT(ABS(B3-A3);"[hh]:mm");B3-A3)	
15:00	15:10	00:10		
15:00	14:51	-00:09		
Alternative 1				
geplante Ankunftszeit	tatsächliche Ankunft:	Differenz	Formel:	
15:00	14:55	-00:05	=WENN(B9-A9<0;"-"&TEXT(ABS(B9-A9);"[hh]:mm");	
15:00	15:10	+00:10	+"&TEXT(ABS(B9-A9);"[hh]:mm"))	
15:00	14:51	-00:09		
Alternative 2				
geplante Ankunftszeit	tatsächliche Ankunft:	Differenz	Formel:	
15:00	14:55	-5	=(B15-A15)*1440	
15:00	15:10	10		
15:00	14:51	-9		

**Lösung**

In der Spalte A sind ab Zeile 3 die geplanten Ankunftszeiten eingetragen. In der Spalte B erfassen Sie ab Zeile 3 die tatsächliche Ankunftszeit. In die Zelle D3 geben Sie folgende Formel ein:

```
=WENN(B3-A3<0;"-"&TEXT(ABS(B3-A3);"[hh]:mm");B3-A3)
```

Als Ergebnis erhalten Sie die Zeitdifferenz als Textwert; hier: »-00:05«.

**Alternative:** Geben Sie in eine Zelle folgende Formel ein:

```
=(B15-A15)*1440
```

Als Ergebnis erhalten Sie die Zeitdifferenz als serielle Zahl; hier: »-5«. Kopieren Sie die Formeln in die übrigen Ergebniszellen.

### Hintergrund

Im ersten Fall werden die positiven Zeitergebnisse als Zeitwerte und die negativen Zeitwerte als Textwerte ausgegeben (vgl. Ausrichtung der Textwerte und der Zahlenwerte!).

Als Ergebnis der Alternative erhalten Sie die Zeitergebnisse als rechenbare Minutewerte (1 Tag = 24 Stunden \* 60 Minuten = 1440 Minuten). Diese Darstellung eignet sich besonders, wenn Sie mit den ausgewiesenen Ergebnissen weiterrechnen müssen.

### Verwendete Funktionen

ABS(Zahl)

TEXT(Zahl;Format)

WENN(Prüfung;DannWert;SonstWert)

## 1.1.7 Berechnung der Arbeitszeit mit Pausenzeit nach sechs Stunden

### Problem

Bei der Berechnung der Arbeitszeiten wird eine Pause von 15 Minuten angerechnet, wenn die Arbeitszeit mehr als 6 Stunden beträgt. Wie errechnen Sie die tatsächlich geleistete Arbeitszeit?

B6		fx Σ v =		=WENN(ODER(B3="";B4="");"";(B4+(B4<B3)-B3)-(WENN((B4+(B4<B3)-B3)>0,25;B5;0)))		
	A	B	C	D		
1	<b>Berechnung der Arbeitszeit mit Pausenzeit</b>					
2						
3	Beginn der Arbeitszeit:	19:00				
4	Ende der Arbeitszeit:	02:00				
5	Pausenzeit bei mehr als 6 Std. Arbeitszeit:	00:15	Formel:			
6	Geleistete Arbeitszeit:	06:45	=WENN(ODER(B3="";B4="");"";(B4+(B4<B3)-B3)-(WENN((B4+(B4<B3)-B3)>0,25;B5;0)))			
7						
8						

### Lösung

Schreiben Sie in die Zelle B3 den Beginn der Arbeitszeit 19:00, in die Zelle B4 das Ende der Arbeitszeit 02:00 und in die Zelle B5 die anzurechnende Pause 0:15 (wenn die Arbeitszeit länger als 6 Stunden ist).

Die anzurechnende Arbeitszeit erhalten Sie in die Zelle C6 mit folgender Formel:

```
=WENN(ODER(B3="";B4="");"";(B4+(B4<B3)-B3)-(WENN((B4+(B4<B3)-B3)>0,25;B5;0)))
```

Als Ergebnis erhalten Sie 06:45 Stunden.



## Hintergrund

Im ersten Teil der Formel wird durch eine WENN()-Funktion eine Fehlermeldung unterdrückt, falls keine Arbeitszeiten eingetragen wurden (ODER(B3="" ; B4="" ) ; ""), sowie die eigentliche Arbeitszeit berechnet (=B4+(B4<B3)-B3). Die Formel berücksichtigt dabei auch einen eventuellen Tageswechsel der Uhrzeiten (B4<B3). Von der berechneten Stundenzahl wird anschließend die Pausenzeit abgezogen, falls die gesamte Arbeitszeit mehr als 6 Stunden (>0,25) ist.

### Verwendete Funktionen

ODER(Wahrheitswert1[;Wahrheitswert2[;...[;Wahrheitswert 255]]])  
 WENN(Prüfung;Dann\_Wert;[Sonst\_Wert])

## 1.1.8 In Industrieeinheiten (60 min = 100 Einheiten) erfasste Arbeitszeit in Stunden und Minuten umwandeln

### Problem

Wie können Sie die auf Ihrem Arbeitsblatt in 100er-Industrie-Zeiteinheiten ausgewiesene Arbeitszeit in Stunden und Minuten umrechnen?

B3	=ZEIT(GANZZAHL(A3/100);REST(A3*0,6;60);0)		
	A	B	C
1	Industriezeiten (60 min = 100 Einheiten) in Stunden und Minuten umrechnen.		
2	Industriezeit	Stunden und Minuten	Formel:
3	275	02:45:00	=ZEIT(GANZZAHL(A3/100);REST(A3*0,6;60);0)
4	100	01:00:00	

### Lösung

In die Zelle A3 schreiben Sie die ausgewiesene Zeiteinheit, z.B. 275.

In die Zelle B3 schreiben Sie die Formel:

```
=ZEIT(GANZZAHL(A3/100);REST(A3*0,6;60);0)
```

Als Ergebnis erhalten Sie die Zeitangabe: »02:45:00«; (2 Stunden, 45 Minuten)

### Hintergrund

Die Formel errechnet zuerst mithilfe der Funktion GANZZAHL() die Stunden, die sich aus dem in Dezimalzahl übergebenen Wert ergeben, und anschließend mithilfe der Funktion REST() die verbleibenden Minuten. Die so gewonnenen Daten werden zum Schluss mithilfe der Funktion ZEIT() in den aktuellen Zeitwert umgewandelt.

### Verwendete Funktionen

GANZZAHL(Zahl)

REST(Dividend;Divisor)

ZEIT(Stunde;Minute;Sekunde)

## 1.2 Tagesberechnungen

### 1.2.1 Gestern – Heute – Morgen

#### Problem

In einer Zelle eines Formulars soll immer der aktuelle Tag angezeigt werden.

Gestern – Heute – Morgen			
			Formel:
Heute:	24.08.22		=HEUTE()
	Heute ist der 24.08.2022		= "Heute ist der "&TEXT(HEUTE());"TT.MM.JJJJ")
Gestern:	23.08.22		=HEUTE()-1
Morgen:	25.08.22		=HEUTE()+1
	Morgen ist der 25.08.2022		= "Morgen ist der "&TEXT(HEUTE()+1);"TT.MM.JJJJ")
Übermorgen:	26.08.22		=HEUTE()+2

#### Lösung

Schreiben Sie die Formel

```
=HEUTE()
```

in die Zelle, die das aktuelle Datum anzeigen soll.

#### Hintergrund

Die Funktion HEUTE() liefert das aktuelle Datum. Die Funktion wird bei jeder Änderung der Tabelle neu berechnet. Diese Funktion hat keine Parameter.

Eingebettet in die Funktion TEXT() kann das Datum auch formatiert und mit Text versehen werden. Das Ergebnis »Heute ist der 06.04.2022« erhalten Sie beispielsweise mit der Formel:

```
= "Heute ist der "&TEXT(HEUTE());"TT.MM.JJJJ")
```

Natürlich können Sie sich auch den gestrigen Tag anzeigen lassen. Die Formel lautet in diesem Falle: = HEUTE()-1, für den morgigen Tag rechnen Sie einfach einen Tag hinzu = HEUTE()+1.

#### Verwendete Funktionen

TEXT(Wert;Format)

HEUTE()

## 1.2.2 Welches Datum hat der nächste Freitag?

### Problem

Die bestellten Waren werden immer an einem bestimmten Tag, hier am nächsten Freitag abgeholt. An welchem Datum erfolgt die nächste Abholung?

B3			
fx Σ = =A3+(WOCHENTAG(A3;2)>5)*7+(5-WOCHENTAG(A3;2))			
	A	B	C
1	<b>Welches Datum hat der nächste Freitag</b>		
2	<b>Datum:</b>	<b>nächster Freitag:</b>	<b>Formeln:</b>
3	Mi, 6. Apr 2022	Fr, 08. Apr 2022	=A3+(WOCHENTAG(A3;2)>5)*7+(5-WOCHENTAG(A3;2))
4	Di, 19. Apr 2022	Fr, 22. Apr 2022	=A4+(WOCHENTAG(A4;2)>5)*7+(5-WOCHENTAG(A4;2))
5	Mi, 20. Jul 2022	Fr, 22. Jul 2022	=A5+(WOCHENTAG(A5;2)>5)*7+(5-WOCHENTAG(A5;2))
6	Fr, 22. Apr 2022	Fr, 22. Apr 2022	=A6+(WOCHENTAG(A6;2)>5)*7+(5-WOCHENTAG(A6;2))
7	<b>Am Freitag bereits den nächsten Freitag anzeigen</b>		
8	Fr, 22. Apr 2022	Fr, 29. Apr 2022	=A8+(WOCHENTAG(A8;2)>=5)*7+(5-WOCHENTAG(A8;2))

### Lösung

Schreiben Sie in die Zelle A3 das aktuelle Datum: 06.04.2022 (= Mittwoch).  
In die Zelle B3 schreiben Sie folgende Formel:

```
=A3+(WOCHENTAG(A3;2)>5)*7+(5-WOCHENTAG(A3;2))
```

Als Ergebnis erhalten Sie: »08.04.2022«.

### Hintergrund

Zum Einsatz kommt die Funktion WOCHENTAG().

Das Argument Zahl ergibt eine ganze Zahl von 1 bis 7. Wird das Argument Art nicht angegeben, hat der Sonntag den Wert 1, hat das Argument Art den Wert 2 oder 11, erhält den Wert 1 der Montag.

Ausgehend vom Datum in Zelle A3 prüft die Formel im ersten Teil, ob der Freitag der laufenden Woche bereits verstrichen ist. Je nach zurückgeliefertem Wahrheitswert Wochentag()>5 liefert die Funktion den Wert 0 (=FALSCH), wenn der Freitag noch bevorsteht, oder 1 (=WAHR), wenn der Freitag bereits verstrichen ist. Ist der Freitag bereits verstrichen, wird zur Wochentagszahl der Wert 1\*7 (= 7 Tage oder 1 Woche) hinzugerechnet und somit der entsprechende Wochentag der nächsten Woche als Ausgangswert zugrunde gelegt.

Der zweite Teil der Formel berechnet die fehlenden Tage bis zum nächsten Freitag. Ergab die Wahrheitsprüfung im ersten Teil der Formel den Wert WAHR, ergibt sich hier eine negative Zahl, sodass als Ergebnis wieder der nächste Freitag erscheint. Im Ergebnis werden zum 06.04.2022 zwei Tage hinzugerechnet; das entsprechende Datum ist Freitag, der 08.04.2022. Soll am Freitag bereits der nächste Freitag angezeigt werden, so wandeln Sie die Formel wie folgt ab:

```
=A3+(WOCHENTAG(A3;2)>=5)*7+(5-WOCHENTAG(A3;2))
```

## Verwendete Funktionen

WENN(Prüfung[; [DannWert] [; [SonstWert]]])

WOCHENTAG(Zahl [; Art])

### 1.2.3 Welches Datum hat der nächste »Freitag, der 13.«?

#### Problem

Alle Terminvereinbarungen, die auf einen »Freitag, den 13.« entfallen, sollen vermieden werden. Wie können Sie solche »Unglücksfreitage« errechnen lassen?

A3		fx Σ v =		{=KKLEINSTE(WENN(((WOCHENTAG(HEUTE()+ZEILE(\$1:2000);2)=5)*(TAG(HEUTE()+ZEILE(\$1:2000))=13))=1;HEUTE()+ZEILE(\$1:2000));ZEILE()-2)}	
A		B			
1	Welches Datum hat der nächste „Freitag der 13.“				
2	Datum:	Formel der Zelle A3:			
3	Fr, 13.05. 22	{=KKLEINSTE(WENN(((WOCHENTAG(HEUTE()+ZEILE(\$1:2000);2)=5)*(TAG(HEUTE()+ZEILE(\$1:2000))=13))=1;HEUTE()+ZEILE(\$1:2000));ZEILE()-2)}			
4	Fr, 13.01. 23				
5	Fr, 13.10. 23				
6	Fr, 13.09. 24				

#### Lösung

Den nächsten Freitag, der auf einen 13. fällt, erhalten Sie beispielsweise in der Zelle A3 mit der Matrixformel:

```
=KKLEINSTE(WENN(((WOCHENTAG(HEUTE()+ZEILE($1:$2000);2)=5)*(TAG(HEUTE()+ZEILE($1:$2000))=13))=1;HEUTE()+ZEILE($1:$2000));ZEILE()-2)
```

Beenden Sie die Eingabe mit **[Strg] + [Shift] + [Enter]**. Als Ergebnis erhalten Sie den nächsten Freitag, der auf einen 13. Montagstag entfällt.

#### Hintergrund

Der Formelteil `WENN(((WOCHENTAG(HEUTE()+ZEILE($1:$2000);2)=5)*(TAG(HEUTE()+ZEILE($1:$2000))=13))=1` prüft durch eine `WENN()`-Funktion ausgehend vom heutigen Tag `=HEUTE()` mit dem Zähler `ZEILE($1:$2000)`, wann in den nächsten 2000 Tagen ein Datum auf den Wochentag 5 (= Freitag) und zugleich auf den 13. Montagstag entfällt. Treffen beide Voraussetzungen für eine Zeilennummer zu, ergibt sich der Wahrheitswert 1 und die `WENN()`-Funktion addiert zum aktuellen Tag die Zahl der gefundenen Zeilennummer.

Das Ergebnis der Matrix ist eine Liste aller gefundenen Freitage, die auf den 13. eines Monats entfallen. Die umschließende Funktion `KKLEINSTE()` extrahiert daraus mithilfe des Parameters `RangC` mit dem Wert `ZEILE()` das kleinste Datum, das zugleich dem nächsten Freitag den 13. entspricht. Nachdem sich die Formel in Zeile 3 befindet, muss der Zähler noch um -2 berichtigt werden.

Um weitere Freitage, die auf einen 13. fallen, angezeigt zu erhalten, kopieren Sie die Matrixformel und fügen sie in die darunter liegenden Zellen ein. Die Funktion KLEINSTE() errechnet automatisch mit dem Zähler ZEILE()-2 die folgenden Datumswerte der nächsten 2000 Tage.

**Verwendete Funktionen**

HEUTE()

KLEINSTE(Daten;RangC)

TAG(Zahl)

WENN(Prüfung;DannWert;SonstWert)

WOCHENTAG(Zahl[;Art])

ZEILE(Bezug)

**1.2.4 Der wievielte Freitag im Monat ist der 12.08.2022?**

**Problem**

Jeden 3. Freitag im Monat wird beispielsweise das Altpapier abgeholt. Wie können Sie ermitteln, wie oft der Freitag in diesem Monat bereits verstrichen ist?

Wie viele Freitage im Monat sind bereits verstrichen?					
Datum:	Woche im Monat:	Freitag des Monats:	Altpapier wird abgeholt	Formeln:	
Mo, 01. Aug 22	1				
Di, 02. Aug 22	1			<b>Formel in B3:</b>	
Mi, 03. Aug 22	1			=ABRUNDEN((A3-DATUM(JAHR(A3);MONAT(A3);1))/7;0)+1	
Do, 04. Aug 22	1				
Fr, 05. Aug 22	1	1		<b>Formel in C3:</b>	
Sa, 06. Aug 22	1			=WENN(WOCHENTAG(A3;11)=5;ABRUNDEN((A3-DATUM(JAHR(A3);MONAT(A3);1))/7;0)+1;"")	
So, 07. Aug 22	1				
Mo, 08. Aug 22	2			<b>Formel in D3:</b>	
Di, 09. Aug 22	2			=WENN(WENN(WOCHENTAG(A3;11)=5;ABRUNDEN((A3-DATUM(JAHR(A3);MONAT(A3);1))/7;0)+1;"")=3;"Altpapier!";"")	
Mi, 10. Aug 22	2				
Do, 11. Aug 22	2				
Fr, 12. Aug 22	2	2			
Sa, 13. Aug 22	2				
So, 14. Aug 22	2				
Mo, 15. Aug 22	3				
Di, 16. Aug 22	3				
Mi, 17. Aug 22	3				
Do, 18. Aug 22	3				
Fr, 19. Aug 22	3	3	Altpapier!		
Sa, 20. Aug 22	3				

**Lösung**

In A14 steht das Datum 12.08.2022.

In B14 schreiben Sie die Formel:

=ABRUNDEN((A14-DATUM(JAHR(A14);MONAT(A14);1))/7;0)+1

Als Ergebnis erhalten Sie die Zahl 2. Das Altpapier wird also am nächsten Freitag abgeholt.

### Hintergrund

Zuerst wird die Anzahl der Tage zwischen dem Datum in A2 und dem ersten Tag des entsprechenden Monats errechnet.

Die Anzahl der errechneten Tage wird durch 7 geteilt (entspricht 7 Tagen = 1 Woche) und auf eine Ganzzahl abgerundet. Nachdem die ersten Tage der Woche aufgrund der Berechnungsmethoden von Calc und Excel den Wert 0 erhalten, müssen Sie zum Ergebnis der Berechnung den Wert 1 hinzurechnen. Als Ergebnis der obigen Prüfung erhalten Sie den Wert 2. Der 12.8.2022 liegt in der Woche mit dem 2. Freitag des Monats. Das Altpapier wird am nächsten Freitag abgeholt.

### Tipp

Sollen die Wochennummern des Monats nur neben jedem Freitag angezeigt werden, ändern Sie die Formel ab in:

```
=WENN(WOCHENTAG(A3;11)=5;ABRUNDEN((A3-DATUM(JAHR(A3);MONAT(A3);1))/7;0)+1;"")
```

Die Wochennummern werden jetzt nur angezeigt, wenn der Wochentag gleich 5 = Freitag ist.

Mit folgender Formel können Sie sich sogar einen Hinweis auf die Altpapierleerung anzeigen lassen:

```
=WENN((WENN(WOCHENTAG(A3;11)=5;ABRUNDEN((A3-DATUM(JAHR(A3);MONAT(A3);1))/7;0)+1;"")=3);"Altpapier!";"")
```

### Verwendete Funktionen

ABRUNDEN(Zahl[;Stellen])

DATUM(Jahr;Monat;Tag)

JAHR(Zahl)

WOCHENTAG(Zahl[;Art])

MONAT(Zahl)

## 1.2.5 Den wievielten Tag des Jahres haben wir heute?

### Problem

Sie möchten wissen, der wievielte Tag des Jahres heute ist?

Den wievielten Tag des Jahres haben wir heute?		
Datum:	Tag:	Formeln:
Jahresanfang:	10	=HEUTE()-DATUM(JAHR(HEUTE());1;1)+1
01.01.2023	10	=HEUTE()-DATUM(JAHR(HEUTE());1;0)
	10	=DATEDIF(A4-1;HEUTE();"d")
=HEUTE()	10	=DATEDIF(A4;HEUTE();"d")+1
10.01.2023	10	=DATEDIF("1.1.2023"-1;HEUTE();"d")
	10	=TAGE(HEUTE();A4)+1
		<b>Jahresanfang</b>
	01.01.23	=DATUM(JAHR(HEUTE());1;1)
	31.12.22	=DATUM(JAHR(HEUTE());1;0)
	01.01.23	=DATUM(JAHR(HEUTE());1;0)+1

### Lösung

Benutzen Sie die Formel:

```
=HEUTE()-DATUM(JAHR(HEUTE());1;1)+1
```

Alternativ stehen Ihnen in Excel die Funktion DATEDIF(), in Calc die Funktion DATUMDIF() zur Verfügung. Jede dieser Formeln liefert Ihnen die aktuelle Tageszahl des entsprechenden Kalenderjahres.

So liefert Ihnen die Formel

```
=DATEDIF(A3-1;HEUTE();"d")
```

die aktuelle Tageszahl des in der Zelle A3 eingetragenen Kalenderjahres.

### Hintergrund

Mit der Formel »Tageszahl von Heute« minus »Tageszahl vom 01.01. des Jahres« wird die Anzahl der dazwischen liegenden Tage berechnet. Beachten Sie dabei, dass der 1. Januar den Wert 0 hat (1-1=0). Sie müssen deshalb einen Tag hinzuaddieren.

Alternativ dazu können Sie die Funktion TAGE(Datum2; Datum1) verwenden. Das Ergebnis liefert die Anzahl der Tage zwischen den beiden Daten. Datum1 ist das Anfangsdatum, Datum2 ist das End-Datum. In beiden Fällen erhalten Sie das richtige Ergebnis.

### Verwendete Funktionen

DATUM(Jahr; Monat; Tag)

DATEDIF(Ausgangsdatum; End-Datum; Einheit) (nur Excel)

DATUMDIF(Start-Datum; End-Datum; Intervall) (nur Calc)

HEUTE()

JAHR(Zahl)

TAGE(Datum2; Datum1)

## 1.2.6 Wie viele Tage sind es bis zum Jahresende?

### Problem

Wie berechnen Sie die tatsächlich verbleibenden Tage bis zum Jahresende?

Wie viele Tage sind es bis zum Jahresende?			
2	Datum:	Tage:	Formel:
3		Verbleibende Tage im Jahr:	
4	20.08.22	133	=DATUM(JAHR(A4);12;31)-A4
5		133	=TAGE("31.12."&JAHR(HEUTE());HEUTE())
6		133	=DATEDIF(HEUTE();"31.12.2022";"d")
8		Anzahl der Tage eines Jahres	
9	01.01.22	365	=DATUM(JAHR(A9)+1;1;1)-DATUM(JAHR(A9);1;1)
10	01.01.24	366	=DATUM(JAHR(A10)+1;1;1)-DATUM(JAHR(A10);1;1)

### Lösung

Die nachfolgenden Formeln können Sie dazu in Excel verwenden:

```
=DATUM(JAHR(HEUTE());12;31)-HEUTE()
```

oder

```
=DATEDIF(HEUTE();"31.12.2022";"d")
```

In jedem Fall ziehen Sie vom Zieldatum 31.12. das Ausgangsdatum ab und berechnen so die Tagesdifferenz.

Verwenden Sie Calc, nehmen Sie die Formel

```
=TAGEIMJAHR(HEUTE())-(HEUTE()-DATUM(JAHR(HEUTE());1;0))
```

oder alternativ =DATUMDIF(HEUTE();"31.12.2022";"d"). Als Ergebnis erhalten Sie die verbleibenden Tage des aktuellen Kalenderjahres.

### Hintergrund

Die Funktion TAGEIMJAHR() berechnet die Anzahl der Kalendertage des Jahres, in das das im Parameter Datum angegebene Datum fällt. Im Kalenderjahr 2022 ergibt sich die Zahl 365. Von dieser Zahl zieht man die bereits vergangenen Tage des aktuellen Kalenderjahres ab.

Alternativ dazu können Sie die verbleibenden Tage eines Kalenderjahres auch mit der Funktion DATEDIF() (in Calc: DATUMDIF()) errechnen. In dieser Funktion werden die Tage zwischen dem Start- und dem End-Datum durch Angabe des 3. Parameters Intervall mit dem Wert d (=Anzahl der Tage) errechnet.



**Verwendete Funktionen**

DATUM(Jahr;Monat;Tag)

HEUTE()

JAHR(Zahl)

TAGE(Datum2;Datum1)

TAGEIMJAHR(Datum)

DATEDIF(Start-Datum;End-Datum;Intervall)

DATUMDIF(Start-Datum;End-Datum;Intervall)

**1.2.7 Individuelle Netto-Arbeitstage**

**Problem**

Das Rechnen mit Netto-Arbeitstagen ist grundsätzlich kein Problem, aber wie berechnen Sie die Arbeitstage, wenn Sie nur an bestimmten Tagen in der Woche, beispielsweise generell nur dienstags, mittwochs und donnerstags zur Arbeit gehen und zudem bestehende Feiertage berücksichtigt werden sollen?

Individuelle Netto-Arbeitstage zwischen Start- und Enddatum												
Startdatum:	01.01.2023	Wochenarbeitstage:							Feiertage:			
Enddatum:	31.01.2023	Arbeitstag = 0 Freier Tag = 1							Name	Datum:		
Name:	Arbeitstage	Mo	Di	Mi	Do	Fr	Sa	So				
Bauer Emil	12	1	0	0	0	1	1	1	1. Weihnachtstag	24.12.22		
Braun Erika	21	0	0	0	0	0	1	1	2. Weihnachtstag	26.12.22		
Schmidt Paul	21								Silvester	31.12.22		
Hausner Berta	11	1	1	0	0	0	1	1	Neujahr	01.01.22		
Pirk Paula	8	1	0	1	0	1	1	1	Dreikönig	06.01.22		
Mayer Sven	12	1	0	0	0	1	1	1	Geburtstag im Monat	19.01.23		
Roth Judit	20	1	0	0	0	0	0	1				
Bauer Hans	11	1	1	0	0	0	1	1				

**Lösung**

Die Funktion NETTOARBEITSTAGE.INTL() errechnet, wie viele Tage jemand in einem bestimmten Zeitraum gearbeitet hat. Dazu besitzt die Funktion einen zusätzlichen Parameter, der bestimmt, welche Tage als freie Tage gewertet werden sollen. Die Funktion rechnet also nicht nur Samstage und Sonntage aus einem Datumsbereich heraus, sondern kann auch andere Tagesbereiche als freie Tage werten.

Steht beispielsweise das Start-Datum in der Zelle B2, das End-Datum in der Zelle B3, die freien Tage im Bereich K4:K50 und arbeiten Sie beispielsweise immer

dienstags, mittwochs und donnerstags, so schreiben Sie in die Berechnungszelle der Arbeitstage folgende Formel:

```
=NETTOARBEITSTAGE.INTL($B$2;$B$3;"1000111";$K$4:$K$50)
```

Beenden Sie die Formel mit der -Taste, berechnet die Formel die tatsächlich geleisteten Arbeitstage zwischen Start- und End-Datum.

### Hintergrund

Die Parameter **Start-Datum** und **End-Datum** sind die Daten, zwischen denen Sie die tatsächlich gearbeiteten Tage berechnen wollen. **Urlaubstage** ist ein Zellbereich, in dem Sie die Datumswerte der Feiertage oder sonstigen freien Tage eingegeben haben, im obigen Beispiel der Bereich K4:K50.

Besonders interessant ist aber der Parameter **Wochenende**. **Wochenende** ist eine optionale Matrix mit Zahlen, die den Standard für eine Arbeitswoche festlegt. Diese Matrix startet mit Montag, Arbeitstage werden durch eine »0«, Nichtarbeitstage durch eine »1« dargestellt. Um das obige Beispiel zu definieren, wo jemand nur dienstags, mittwochs und donnerstags arbeitet, geben Sie folgende Zeichenfolge ein: "1000111".

Noch variabler wird die Berechnung, wenn Sie die Werte für den Parameter **Wochenende** ebenfalls in einzelne Tabellenzellen eingeben – zum Beispiel in die Zelle C5 den Parameter für Montag, in die Zelle D5 den Parameter für Dienstag, in E5 den Parameter für Mittwoch bis in die Zelle I5 den Parameter für Sonntag. In die einzelnen Zellen tragen Sie dann 0 für einen Arbeitstag oder 1 für einen freien Tag ein.

Die Formel zur Berechnung der Arbeitstage des in der Zeile 5 eingetragenen Mitarbeiters setzt sich schließlich so zusammen:

```
=NETTOARBEITSTAGE.INTL($B$2;$B$3;TEXTKETTE($C5:$I5);$K$4:$K$51)
```

So kann für jeden Mitarbeiter mithilfe der Funktion **TEXTKETTE()** als dritter Parameter **Wochenende** die Arbeitszeit ganz individuell berechnet werden. Sie brauchen die Formel nur nach unten zu kopieren.

### Verwendete Funktionen

**NETTOARBEITSTAGE**(Ausgangsdatum;End-Datum;[Freie\_Tage]) (nur Excel)

**NETTOARBEITSTAGE**(Start-Datum;End-Datum[;[Freie\_Tage];

[Wochenende]]) (nur Calc)

**NETTOARBEITSTAGE.INTL**(Start-Datum;End-Datum[;

[Wochenende];[Urlaubstage]])

**TEXTKETTE**(Zeichenkette1[;Zeichenkette2[;...[;Zeichenkette255]]])